Pensieve Header: The Burau representation in β-calculus.

The first two sections are borrowed from betaCalculus.nb; the third largely from 2012-01/betaCalculus.nb.

In[51]:=
```
<< KnotTheory`
GD[K_] := GD @@ (
   PD[K] /. X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]],
      Ar[l, i, +1], Ar[j, i, -1]
     ]
  )
```

```
Loading KnotTheory` version of August 22, 2010, 13:36:57.55.
Read more at http://katlas.org/wiki/KnotTheory.
```

Utilities

In[1]:=
```
βSimplify = Factor;
SetAttributes[βCollect, Listable];
βCollect[B[ω_, μ_]] := B[
   βSimplify[ω],
   Collect[μ, _h, Collect[#, _t, βSimplify] &]
  ];
(* "L" for "Labels" *)
hL[β_] := Union[Cases[β, h[s_] :> s, Infinity]];
tL[β_] := Union[Cases[β, t[s_] | c_s_ :> s, Infinity]];
dL[β_] := Union[hL[β], tL[β]];
βForm[B[ω_, μ_]] := Module[
   {tails, heads, mat},
   tails = tL[B[ω, μ]]; heads = hL[B[ω, μ]];
   mat = Outer[βSimplify[Coefficient[μ, h[#1] t[#2]]] &, heads, tails];
   PrependTo[mat, t /@ tails];
   mat = Prepend[Transpose[mat], Prepend[h /@ heads, ω]];
   MatrixForm[mat]
  ];
βForm[else_] := else /. β_B :> βForm[β];
B /: B[ω1_, β1_] == B[ω2_, β2_] := (ω1 == ω2) && (β1 == β2);
```

The Meta-Cross-Product

The “Tails” meta-group

In[10]:=
```
tm[x_, y_, z_][β_] := β /. {t[x] → t[z], t[y] → t[z], c_x → c_z, c_y → c_z};
tΔ[z_, x_, y_][β_] := βCollect[β /. {t[z] → t[x] + t[y], c_z → c_x + c_y}];
tη[x_][β_] := βCollect[(β /. t[x] → 0) /. c_x → 0];
tS[x_][β_] := βCollect[β /. {t[x] → -t[x], c_x → -c_x}];
tA[_][β_] := βCollect[β];
tP[rules___Rule][β_] := βCollect[
   β /. {t[x_] :> t[x /. {rules}], c_x_ :> c_(x /. {rules})}
  ];
```

The “Heads” meta-group

```
In[16]:= hm[x_, y_, z_][B[ω_, μ_]] := Module[
           {γx = D[μ, h[x]], γy = D[μ, h[y]], M = μ /. h[x] | h[y] → 0},
           B[ω, M + h[z] (γx + γy + (γx /. t[i_] ↦ c_i) γy)] // βCollect
          ];
         hΔ[z_, x_, y_][β_] := βCollect[β /. h[z] → h[x] + h[y]];
         hη[x_][β_] := βCollect[β /. h[x] → 0];
         hS[x_][B[ω_, μ_]] := Module[{γ},
           γ = 1 + D[μ, h[x]] /. t[s_] ↦ c_s;
           βCollect[B[ω, μ /. h[x] → -h[x] / γ]]
          ];
         hA[x_][β_] := hS[x][β];
         hP[rules___Rule][β_] := βCollect[β /. h[x_] ↦ h[x /. {rules}]];
```

The TH → HT Swap

```
In[22]:= thswap[x_, y_][B[ω_, μ_]] := Module[
           {α, β, γ, δ, ϵ},
           α = Coefficient[μ, h[x] t[y]];
           β = D[μ, t[y]] /. h[x] → 0;
           γ = D[μ, h[x]] /. t[y] → 0;
           δ = μ /. h[x] | t[y] → 0;
           ϵ = 1 + c_y α;
           B[ω * ϵ, Plus[
              α (1 + (γ /. t[i_] ↦ c_i) / ϵ) h[x] t[y],
              β (1 + (γ /. t[i_] ↦ c_i) / ϵ) t[y],
              γ / ϵ h[x],
              δ - c_y / ϵ γ * β
             ]] // βCollect
          ];
```

The “double” meta-group

```
In[23]:= dm[x_, y_, z_][β_] := β // thswap[y, x] // hm[x, y, z] // tm[x, y, z];
```

The “external” product

```
In[24]:= B /: B[ω1_, μ1_] B[ω2_, μ2_] := B[ω1 * ω2, μ1 + μ2];
```

The R-Matrix

```
In[25]:= R[x_, y_] := B[1, (E^c_x - 1) / c_x * t[x] h[y]];
         Rinv[x_, y_] := B[1, (E^(-c_x) - 1) / c_x * t[x] h[y]];
```

The Group A_n

```
In[27]:= ar[i_, j_] := t[i] h[j];
         htswap[y_, x_][β_] := β // hS[x] // thswap[y, x] // hS[x];
         dA[x_][β_] := β // tA[x] // htswap[x, x] // hA[x];
         dS[x_][β_] := β // tS[x] // htswap[x, x] // hS[x];
```

```
In[31]:= Unprotect[NonCommutativeMultiply];
β_ ** ν_ := Module[
   {ρ, σ, labels},
   ρ = β * (ν /. {h[s_] :> h[σ[s]], t[s_] :> t[σ[s]], c_s_ :> c_σ[s]});
   labels = Union[Cases[{β, ν}, h[s_] | t[s_] | c_s_ :> s, Infinity]];
   Do[
    ρ = ρ // dm[s, σ[s], s],
    {s, labels}
   ];
   ρ
  ];
B /: Inverse[B[ω_, μ_]] := Module[
   {ρ = B[1, μ]},
   Do[ρ = ρ // dA[s], {s, Union[hL[ρ], tL[ρ]]}];
   ReplacePart[ρ, 1 → 1 / ω] // βCollect
  ];
```

The Burau Representation

```
In[34]:= Burau[B[_, μ_]] := Module[
   {labels, ν},
   labels = Union[hL[μ], tL[μ]];
   ν = (μ /. h[j_] :> -c_j h[j]) + (μ /. t[i_] :> c_i /. h[j_] :> t[j] h[j]);
   Outer[Coefficient[ν, t[#2] h[#1]] &, labels, labels] +
    IdentityMatrix[Length[labels]]
  ];
Burau[n_, B[_, μ_]] := Module[
   {labels, ν},
   labels = Range[n];
   ν = (μ /. h[j_] :> -c_j h[j]) + (μ /. t[i_] :> c_i /. h[j_] :> t[j] h[j]);
   Outer[Coefficient[ν, t[#2] h[#1]] &, labels, labels] + IdentityMatrix[n]
  ];
```

Testing

```
In[36]:= {
  R[1, 2] ** R[1, 3] ** R[2, 3],
  R[2, 3] ** R[1, 3] ** R[1, 2]
 } // βForm
```

Out[36]=

$$\left\{\begin{pmatrix} 1 & h[2] & h[3] \\ t[1] & \frac{-1+e^{c_1}}{c_1} & \frac{-1+e^{c_1}}{c_1} \\ t[2] & 0 & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix}, \begin{pmatrix} 1 & h[2] & h[3] \\ t[1] & \frac{-1+e^{c_1}}{c_1} & \frac{-1+e^{c_1}}{c_1} \\ t[2] & 0 & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix}\right\}$$

In[37]:=
```
n = 3;
{
   α1 = B[1, Sum[α10 i+j ar[i, j], {i, n}, {j, n}]],
   β1 = B[1, Sum[β10 i+j ar[i, j], {i, n}, {j, n}]],
   α1 ** β1
  } // βForm // ColumnForm
```

Out[38]=

1	h[1]	h[2]	h[3]
t[1]	α_{11}	α_{12}	α_{13}
t[2]	α_{21}	α_{22}	α_{23}
t[3]	α_{31}	α_{32}	α_{33}

1	h[1]	h[2]	h[3]
t[1]	β_{11}	β_{12}	β_{13}
t[2]	β_{21}	β_{22}	β_{23}
t[3]	β_{31}	β_{32}	β_{33}

1	h[1]	
t[1]	$\alpha_{11} + \beta_{11} + c_1\,\alpha_{11}\,\beta_{11} + c_2\,\alpha_{21}\,\beta_{11} + c_3\,\alpha_{31}\,\beta_{11} - c_2\,\alpha_{21}\,\beta_{12} - c_3\,\alpha_{31}\,\beta_{13} + c_2\,\alpha_{11}\,\beta_{21} + c_3\,\alpha_{11}\,\beta_{31}$	
t[2]	$\alpha_{21} + c_1\,\alpha_{21}\,\beta_{12} + \beta_{21} + c_2\,\alpha_{21}\,\beta_{21} + c_3\,\alpha_{31}\,\beta_{21} - c_3\,\alpha_{31}\,\beta_{23} + c_3\,\alpha_{21}\,\beta_{32}$	$\alpha_{22} + c_1$
t[3]	$\alpha_{31} + c_1\,\alpha_{31}\,\beta_{13} + c_2\,\alpha_{31}\,\beta_{23} + \beta_{31} + c_2\,\alpha_{21}\,\beta_{31} + c_3\,\alpha_{31}\,\beta_{31} - c_2\,\alpha_{21}\,\beta_{32}$	

In[39]:=
```
n = 2;
{
   α1 = B[1, Sum[α10 i+j ar[i, j], {i, n}, {j, n}]],
   β1 = α1 // dS[1] // dS[2],
   α1 ** β1
  } // βForm // ColumnForm
```

Out[40]=

1	h[1]	h[2]
t[1]	α_{11}	α_{12}
t[2]	α_{21}	α_{22}

$-\frac{-1+c_1\,\alpha_{12}+c_2\,\alpha_{21}}{(-1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(-1+c_1\,\alpha_{12}+c_2\,\alpha_{22})}$	h[1]	h[2]
t[1]	$-\frac{-\alpha_{11}+c_1\,\alpha_{11}\,\alpha_{12}+c_2\,\alpha_{12}\,\alpha_{21}}{(-1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(-1+c_1\,\alpha_{12}+c_2\,\alpha_{21})}$	$-\frac{\alpha_{12}}{-1+c_1\,\alpha_{12}+c_2\,\alpha_{21}}$
t[2]	$-\frac{\alpha_{21}}{-1+c_1\,\alpha_{12}+c_2\,\alpha_{21}}$	$-\frac{c_1\,\alpha_{12}\,\alpha_{21}-\alpha_{22}+c_2\,\alpha_{21}\,\alpha_{22}}{(-1+c_1\,\alpha_{12}+c_2\,\alpha_{21})\,(-1+c_1\,\alpha_{12}+c_2\,\alpha_{22})}$

$-\frac{-1+c_1\,\alpha_{12}+c_2\,\alpha_{21}}{(-1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(-1+c_1\,\alpha_{12}+c_2\,\alpha_{22})}$	h[1]	h[2]
t[1]	$-\frac{2\,(-\alpha_{11}+c_1\,\alpha_{11}\,\alpha_{12}+c_2\,\alpha_{12}\,\alpha_{21})}{(-1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(-1+c_1\,\alpha_{12}+c_2\,\alpha_{21})}$	$-\frac{2\,\alpha_{12}}{-1+c_1\,\alpha_{12}+c_2\,\alpha_{21}}$
t[2]	$-\frac{2\,\alpha_{21}}{-1+c_1\,\alpha_{12}+c_2\,\alpha_{21}}$	$-\frac{2\,(c_1\,\alpha_{12}\,\alpha_{21}-\alpha_{22}+c_2\,\alpha_{21}\,\alpha_{22})}{(-1+c_1\,\alpha_{12}+c_2\,\alpha_{21})\,(-1+c_1\,\alpha_{12}+c_2\,\alpha_{22})}$

In[41]:=
```
n = 3;
{
  α1 = B[ω, Sum[α_{10 i+j} ar[i, j], {i, n}, {j, n}]],
  β1 = α1 // dA[1] // dA[2] // dA[3],
  α1 ** β1,
  α1 ** Inverse[α1]
 } // βForm // ColumnForm
```

Out[42]=

ω	h[1]	h[2]	h[3]
t[1]	α_{11}	α_{12}	α_{13}
t[2]	α_{21}	α_{22}	α_{23}
t[3]	α_{31}	α_{32}	α_{33}

$$\frac{\omega\,(1+c_1\alpha_{12}+c_1\alpha_{13}+c_1^2\alpha_{12}\alpha_{13}+c_2\alpha_{21}+c_1c_2\alpha_{13}\alpha_{21}+c_2\alpha_{23}+c_1c_2\alpha_{12}\alpha_{23}+c_2^2\alpha_{21}\alpha_{23}+c_3\alpha_{31}+c_1c_3\alpha_{12}\alpha_{31}+c_2c_3\alpha_{23}\alpha_{31}+c_3\alpha_{32}+c_1c_3\alpha_{13}\alpha_{32}+c_2c_3\alpha_{21}\cdots}{(1+c_1\alpha_{11}+c_2\alpha_{21}+c_3\alpha_{31})\,(1+c_1\alpha_{12}+c_2\alpha_{22}+c_3\alpha_{32})\,(1+c_1\alpha_{13}+c_2\alpha_{23}+c_3\alpha_{33})}$$

t[1]

t[2]

t[3]

$$\frac{\omega^2\,(1+c_1\alpha_{12}+c_1\alpha_{13}+c_1^2\alpha_{12}\alpha_{13}+c_2\alpha_{21}+c_1c_2\alpha_{13}\alpha_{21}+c_2\alpha_{23}+c_1c_2\alpha_{12}\alpha_{23}+c_2^2\alpha_{21}\alpha_{23}+c_3\alpha_{31}+c_1c_3\alpha_{12}\alpha_{31}+c_2c_3\alpha_{23}\alpha_{31}+c_3\alpha_{32}+c_1c_3\alpha_{13}\alpha_{32}+c_2c_3\alpha_{21}\cdots}{(1+c_1\alpha_{11}+c_2\alpha_{21}+c_3\alpha_{31})\,(1+c_1\alpha_{12}+c_2\alpha_{22}+c_3\alpha_{32})\,(1+c_1\alpha_{13}+c_2\alpha_{23}+c_3\alpha_{33})}$$

t[1]
t[2]
t[3]

(1)

In[43]:=
```
{
  α1 = R[1, 2] ** R[2, 3],
  β1 = α1 // dA[1] // dA[2] // dA[3],
  Inverse[α1],
  α1 ** β1
 } // βForm // ColumnForm
```

Out[43]=

1	h[2]	h[3]
t[1]	$\frac{-1+e^{c_1}}{c_1}$	0
t[2]	0	$\frac{-1+e^{c_2}}{c_2}$

1	h[2]	h[3]
t[1]	$-\frac{e^{-c_1}(-1+e^{c_1})}{c_1}$	$-\frac{e^{-c_1-c_2}(-1+e^{c_1})(-1+e^{c_2})}{c_1}$
t[2]	0	$-\frac{e^{-c_1-c_2}(-1+e^{c_2})}{c_2}$

1	h[2]	h[3]
t[1]	$-\frac{e^{-c_1}(-1+e^{c_1})}{c_1}$	$-\frac{e^{-c_1-c_2}(-1+e^{c_1})(-1+e^{c_2})}{c_1}$
t[2]	0	$-\frac{e^{-c_1-c_2}(-1+e^{c_2})}{c_2}$

(1)

In[44]:=
```
n = 3;
{
  α1 = B[ω, Sum[α10 i+j ar[i, j], {i, n}, {j, n}]],
  Inverse[α1] ** α1,
  (α1 ** B[1, ϵ * ar[1, 0]] ** Inverse[α1]) // βCollect,
  Inverse[α1] ** B[1, ϵ * ar[1, 0]] ** α1,
  Inverse[α1] ** B[1, ϵ * ar[2, 0]] ** α1,
  Burau[α1] // MatrixForm
 } // βForm // ColumnForm
```

Out[45]=

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] \\ t[1] & \alpha_{11} & \alpha_{12} & \alpha_{13} \\ t[2] & \alpha_{21} & \alpha_{22} & \alpha_{23} \\ t[3] & \alpha_{31} & \alpha_{32} & \alpha_{33} \end{pmatrix}$$

$$(1)$$

$$\begin{pmatrix} 1 & h[0] \\ t[1] & \frac{\epsilon\,(1+c_1\alpha_{12}+c_1\alpha_{13}+c_1^2\alpha_{12}\alpha_{13}+c_2\alpha_{23}+c_1c_2\alpha_{12}\alpha_{23}+c_3\alpha_{32}+c_1c_3\alpha_{13}\alpha_{32})}{1+c_1\alpha_{12}+c_1\alpha_{13}+c_1^2\alpha_{12}\alpha_{13}+c_2\alpha_{21}+c_1c_2\alpha_{13}\alpha_{21}+c_2\alpha_{23}+c_1c_2\alpha_{12}\alpha_{23}+c_2^2\alpha_{21}\alpha_{23}+c_3\alpha_{31}+c_1c_3\alpha_{12}\alpha_{31}+c_2c_3\alpha_{23}\alpha_{31}+c_3\alpha_{32}+c_1c_3\alpha_{13}\alpha_{32}+c_2\,(} \\ t[2] & \frac{\epsilon\,c_1\,(\alpha_{21}+c_1\alpha_{13}\alpha_{21}+c_2\alpha_{21}\alpha_{23}+c_3\alpha_{23}\alpha_{31})}{1+c_1\alpha_{12}+c_1\alpha_{13}+c_1^2\alpha_{12}\alpha_{13}+c_2\alpha_{21}+c_1c_2\alpha_{13}\alpha_{21}+c_2\alpha_{23}+c_1c_2\alpha_{12}\alpha_{23}+c_2^2\alpha_{21}\alpha_{23}+c_3\alpha_{31}+c_1c_3\alpha_{12}\alpha_{31}+c_2c_3\alpha_{23}\alpha_{31}+c_3\alpha_{32}+c_1c_3\alpha_{13}\alpha_{32}+c_2\,(} \\ t[3] & \frac{\epsilon\,c_1\,(\alpha_{31}+c_1\alpha_{12}\alpha_{31}+c_2\alpha_{21}\alpha_{32}+c_3\alpha_{31}\alpha_{32})}{1+c_1\alpha_{12}+c_1\alpha_{13}+c_1^2\alpha_{12}\alpha_{13}+c_2\alpha_{21}+c_1c_2\alpha_{13}\alpha_{21}+c_2\alpha_{23}+c_1c_2\alpha_{12}\alpha_{23}+c_2^2\alpha_{21}\alpha_{23}+c_3\alpha_{31}+c_1c_3\alpha_{12}\alpha_{31}+c_2c_3\alpha_{23}\alpha_{31}+c_3\alpha_{32}+c_1c_3\alpha_{13}\alpha_{32}+c_2\,(} \end{pmatrix}$$

$$\begin{pmatrix} 1 & h[0] \\ t[1] & \epsilon\,(1+c_2\alpha_{21}+c_3\alpha_{31}) \\ t[2] & -\epsilon\,c_1\alpha_{21} \\ t[3] & -\epsilon\,c_1\alpha_{31} \end{pmatrix}$$

$$\begin{pmatrix} 1 & h[0] \\ t[1] & -\epsilon\,c_2\alpha_{12} \\ t[2] & \epsilon\,(1+c_1\alpha_{12}+c_3\alpha_{32}) \\ t[3] & -\epsilon\,c_2\alpha_{32} \end{pmatrix}$$

$$\begin{pmatrix} 1+c_2\alpha_{21}+c_3\alpha_{31} & -c_1\alpha_{21} & -c_1\alpha_{31} \\ -c_2\alpha_{12} & 1+c_1\alpha_{12}+c_3\alpha_{32} & -c_2\alpha_{32} \\ -c_3\alpha_{13} & -c_3\alpha_{23} & 1+c_1\alpha_{13}+c_2\alpha_{23} \end{pmatrix}$$

In[46]:=
```
n = 3;
{
  α1 = B[ω, Sum[α_{10 i+j} ar[i, j], {i, n}, {j, n}]],
  β1 = B[ω, Sum[β_{10 i+j} ar[i, j], {i, n}, {j, n}]],
  (t1 = Burau[α1]) // MatrixForm,
  (t2 = Burau[β1]) // MatrixForm,
  (t3 = Burau[α1 ** β1]) // Expand // MatrixForm,
  t1.t2 // Expand // MatrixForm,
  Expand[t1.t2 == t3]
 } // βForm // ColumnForm
```

Out[47]=

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] \\ t[1] & \alpha_{11} & \alpha_{12} & \alpha_{13} \\ t[2] & \alpha_{21} & \alpha_{22} & \alpha_{23} \\ t[3] & \alpha_{31} & \alpha_{32} & \alpha_{33} \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] \\ t[1] & \beta_{11} & \beta_{12} & \beta_{13} \\ t[2] & \beta_{21} & \beta_{22} & \beta_{23} \\ t[3] & \beta_{31} & \beta_{32} & \beta_{33} \end{pmatrix}$$

$$\begin{pmatrix} 1 + c_2\,\alpha_{21} + c_3\,\alpha_{31} & -c_1\,\alpha_{21} & -c_1\,\alpha_{31} \\ -c_2\,\alpha_{12} & 1 + c_1\,\alpha_{12} + c_3\,\alpha_{32} & -c_2\,\alpha_{32} \\ -c_3\,\alpha_{13} & -c_3\,\alpha_{23} & 1 + c_1\,\alpha_{13} + c_2\,\alpha_{23} \end{pmatrix}$$

$$\begin{pmatrix} 1 + c_2\,\beta_{21} + c_3\,\beta_{31} & -c_1\,\beta_{21} & -c_1\,\beta_{31} \\ -c_2\,\beta_{12} & 1 + c_1\,\beta_{12} + c_3\,\beta_{32} & -c_2\,\beta_{32} \\ -c_3\,\beta_{13} & -c_3\,\beta_{23} & 1 + c_1\,\beta_{13} + c_2\,\beta_{23} \end{pmatrix}$$

$$\begin{pmatrix} 1 + c_2\,\alpha_{21} + c_3\,\alpha_{31} + c_1\,c_2\,\alpha_{21}\,\beta_{12} + c_1\,c_3\,\alpha_{31}\,\beta_{13} + c_2\,\beta_{21} + c_2^2\,\alpha_{21}\,\beta_{21} + c_2\,c_3\,\alpha_{31}\,\beta_{21} + c_3\,\beta_{31} + c_2\,c_3\,\alpha_{21}\,\beta_{31} + \\ -c_2\,\alpha_{12} - c_2\,\beta_{12} - c_1\,c_2\,\alpha_{12}\,\beta_{12} - c_2\,c_3\,\alpha_{32}\,\beta_{12} + c_2\,c_3\,\alpha_{32}\,\beta_{13} - c_2^2\,\alpha_{12}\,\beta_{21} - c_2\,c_3\,\alpha_{12}\,\beta_{31} \\ -c_3\,\alpha_{13} + c_2\,c_3\,\alpha_{23}\,\beta_{12} - c_3\,\beta_{13} - c_1\,c_3\,\alpha_{13}\,\beta_{13} - c_2\,c_3\,\alpha_{23}\,\beta_{13} - c_2\,c_3\,\alpha_{13}\,\beta_{21} - c_3^2\,\alpha_{13}\,\beta_{31} \end{pmatrix}$$

$$\begin{pmatrix} 1 + c_2\,\alpha_{21} + c_3\,\alpha_{31} + c_1\,c_2\,\alpha_{21}\,\beta_{12} + c_1\,c_3\,\alpha_{31}\,\beta_{13} + c_2\,\beta_{21} + c_2^2\,\alpha_{21}\,\beta_{21} + c_2\,c_3\,\alpha_{31}\,\beta_{21} + c_3\,\beta_{31} + c_2\,c_3\,\alpha_{21}\,\beta_{31} + \\ -c_2\,\alpha_{12} - c_2\,\beta_{12} - c_1\,c_2\,\alpha_{12}\,\beta_{12} - c_2\,c_3\,\alpha_{32}\,\beta_{12} + c_2\,c_3\,\alpha_{32}\,\beta_{13} - c_2^2\,\alpha_{12}\,\beta_{21} - c_2\,c_3\,\alpha_{12}\,\beta_{31} \\ -c_3\,\alpha_{13} + c_2\,c_3\,\alpha_{23}\,\beta_{12} - c_3\,\beta_{13} - c_1\,c_3\,\alpha_{13}\,\beta_{13} - c_2\,c_3\,\alpha_{23}\,\beta_{13} - c_2\,c_3\,\alpha_{13}\,\beta_{21} - c_3^2\,\alpha_{13}\,\beta_{31} \end{pmatrix}$$

True

In[48]:=
```
Burau[5, R[1, 2]] // MatrixForm
```

Out[48]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ -\frac{(-1+e^{c_1})\,c_2}{c_1} & e^{c_1} & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

In[49]:=
```
Burau[5, R[1, 2] ** R[2, 3] ** Rinv[1, 2] ** Rinv[2, 3]] // MatrixForm
```

Out[49]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ -\frac{e^{-c_1}\,(-1+e^{c_1})\,(-1+e^{c_2})\,c_3}{c_1} & \frac{e^{-c_1}\,(-1+e^{c_1})\,(-1+e^{c_2})\,c_3}{c_2} & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

In[50]:=
```
R[1, 2] ** R[2, 3] ** Rinv[1, 2] ** Rinv[2, 3] // βForm
```

Out[50]//MatrixForm=

$$\begin{pmatrix} 1 & h[3] \\ t[1] & \frac{e^{-c_1}(-1+e^{c_1})(-1+e^{c_2})}{c_1} \\ t[2] & -\frac{e^{-c_1}(-1+e^{c_1})(-1+e^{c_2})}{c_2} \end{pmatrix}$$

In[53]:=
```
βMVA[L_Link] := Module[
  {skel, β, s, k},
  skel = Skeleton[L];
  β = Times @@ GD[L] /. {Ar[x_, y_, +1] :> R[x, y],  Ar[x_, y_, -1] :> Rinv[x, y]};
  Do[
   Do[
    β = β // dm[skel[[s, 1]], skel[[s, k]], skel[[s, 1]]],
    {k, 2, Length[skel[[s]]]}
   ],
   {s, Length[skel]}
  ];
  β
 ]
```

In[71]:=
```
MultivariableAlexander[L = Link["L6a5"]][T] /. T[i_] :> T_i
```

Out[71]=

$$\frac{-T_1 - T_2 + T_1 T_2 - T_3 + T_1 T_3 + T_2 T_3}{\sqrt{T_1}\ \sqrt{T_2}\ \sqrt{T_3}}$$

In[69]:=
```
(β0 = βMVA[L]) // βForm
```

Out[69]//MatrixForm=

$$\begin{pmatrix} e^{-2c_1-2c_5-2c_9}(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9}) & h[1] & \cdots \\ t[1] & -\frac{e^{-c_5-c_9}(-1+e^{c_1})(1-e^{c_1}-e^{c_5}-e^{c_9}+e^{c_5+c_9}+e^{c_1+c_5+c_9})}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})\,c_1} & \cdots \\ t[5] & -\frac{(-1+e^{c_5})(-e^{c_1}-e^{c_5}+e^{c_1+c_5}+e^{c_1+c_9}+e^{c_5+c_9})}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9})\,c_5} & -\frac{e^{-c}}{\cdots} \\ t[9] & -\frac{(-1+e^{c_9})(1-e^{c_1}-e^{c_5}+e^{c_1+c_5}-e^{c_9}+e^{c_1+c_9}+e^{c_5+c_9})}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9})\,c_9} & \cdots \end{pmatrix}$$

In[63]:=
```
(β1 = βCollect[β0 /. B[ω_, μ_] :> B[ω, ω * μ]]) // βForm
```

Out[63]//MatrixForm=

$$\begin{pmatrix} e^{-2c_1-2c_5-2c_9}(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9}) & h[1] & \cdots \\ t[1] & -\frac{e^{-2c_1-3c_5-3c_9}(-1+e^{c_1})(-1+e^{c_5}+e^{c_9})(1-e^{c_1}-e^{c_5}-e^{c_9}+e^{c_5+\cdots}}{c_1} & \cdots \\ t[5] & -\frac{e^{-2c_1-2c_5-2c_9}(-1+e^{c_5})(-e^{c_1}-e^{c_5}+e^{c_1+c_5}+e^{c_1+c_9}+e^{c_5\cdots}}{c_5} & \cdots \\ t[9] & -\frac{e^{-2c_1-2c_5-2c_9}(-1+e^{c_9})(1-e^{c_1}-e^{c_5}+e^{c_1+c_5}-e^{c_9}+e^{c_1+c_9}+\cdots}{c_9} & \cdots \end{pmatrix}$$

In[66]:=
```
Collect[β0[[2]] /. t[i_] :> c_i, _h, Simplify]
```

Out[66]= $(-1+e^{-c_5-c_9})\, h[1] + (-1+e^{-c_1-c_9})\, h[5] + (-1+e^{-c_1-c_5})\, h[9]$

In[60]:=
```
Simplify[Burau[β0]] // MatrixForm
```

Out[60]//MatrixForm=

$$\begin{pmatrix} \frac{e^{2c_1}}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})} & \frac{(-1+e^{c_5})(-e^{c_1}-e^{c_5}+e^{c_1+c_5}+e^{c_1+c_9}+e^{c_5+c_9})\,c_1}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9})\,c_5} & \frac{(-1+e^{c_9})(1-e^{c_1}-e^{c_5}+e^{c_1+c_5}-e^{c_9}+e^{c_1+c_9}+e^{c}\cdots}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c}\cdots} \\ \frac{e^{c_1}(-1+e^{c_1})\,c_5}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})\,c_1} & \frac{-1+e^{c_1}+e^{c_5}-e^{2c_5}-e^{c_1+c_5}+e^{c_1+2c_5}+e^{c_9}-e^{c_1+c_9}-e^{c_5+c_9}+e^{c_1+c_5+c_9}+e^{2c_5+c_9}}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9})} & \frac{(-1+e^{c_9})(-e^{c_5}+e^{c_1+c_5}-e^{c_9}+e^{c_1+c_9}+e^{c_5+c}\cdots}{(-1+e^{c_1}+e^{c_5})(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c}\cdots} \\ \frac{(-1+e^{c_1})\,c_9}{(-1+e^{c_1}+e^{c_9})\,c_1} & \frac{e^{c_9}(-1+e^{c_5})\,c_9}{(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9})\,c_5} & \frac{e^{2c_9}}{(-1+e^{c_1}+e^{c_9})(-1+e^{c_5}+e^{c_9})} \end{pmatrix}$$

In[68]:=
```
Simplify[Burau[β0].{c1, c5, c9}]
```
$$\text{Simplify}[\text{Burau}[\beta 0].\{c_1, c_5, c_9\}]$$

Out[68]= $\{c_1, c_5, c_9\}$

In[79]:=
$$\left(e^{-2 c_1 - 2 c_5 - 2 c_9} \left(-1 + e^{c_1} + e^{c_5}\right) \left(-1 + e^{c_1} + e^{c_9}\right) \left(-1 + e^{c_5} + e^{c_9}\right)\right) * \text{Det}[\text{Simplify}[\text{Burau}[\beta 0]]] \text{ // Simplify}$$

Out[79]= $e^{-2 (c_1 + c_5 + c_9)}$